﻿// forked from clockmaker's 重力マウス（プチ軽量化：10万パーティクル）
// forked from coppieee's 重力マウス（さらに軽量化してみた）
// forked from paq's forked from: 重力マウス（ちょっぴり軽量化してみた）
// forked from fumix's 重力マウス（リンクリストにしてみた）
// forked from undo's 重力マウス
//　リンクリストにしてみたけどそんなに速くない？？
//_bmd.fillRect()を_bmd.setPixel()に変更。
//sin(),cos(),atan2(),sqrt()を排除。
// Add final class / mouseEnalbled = false by clockmaker
// CJ Cat: Used vector instead of linked list.
package
{
	import flash.display.Bitmap;
	import flash.display.BitmapData;
	import flash.display.Sprite;
	import flash.display.StageAlign;
	import flash.display.StageScaleMode;
	import flash.events.Event;
	import flash.geom.ColorTransform;
	import flash.geom.Point;
	import flash.geom.Rectangle;
	import flash.utils.ByteArray;
	import net.hires.debug.Stats;

	[SWF(frameRate='60', width='465', height='465', backgroundColor='0x0')]

	import flash.utils.Dictionary;
	public class Main extends Sprite {
		private var _bmp:Bitmap;
		private var _bmd:BitmapData;
		private var _bmdRect:Rectangle;
		private var _colorTransform:ColorTransform = new ColorTransform(0.6, 0.7, 0.96, 1.0);

		private const _maxNum:int = 250000; //50万パーティクルでも45fpsぐらいでる
		private var _cnt:int = 0;
		
		private const _particleData:Vector.<Number> = new Vector.<Number>(_maxNum * 4, true);
		

		public function Main() {
			if(stage) init(null);
			else addEventListener(Event.ADDED_TO_STAGE, init);
			return;
		}
		
		private function init(evt:Event):void {
			removeEventListener(Event.ADDED_TO_STAGE, init);
			
			this.stage.align = StageAlign.TOP_LEFT;
			this.stage.scaleMode = StageScaleMode.NO_SCALE;
			this.stage.quality = "low";
			this.stage.frameRate = 120;
			this.mouseEnabled = false;
			this.mouseChildren = false;
		
			_bmd = new BitmapData(465, 465, false, 0x000000);
			_bmp = new Bitmap(_bmd);
			addChild(_bmp);
			this._bmdRect = _bmd.rect;
			
			var i0:int, i1:int, i2:int, i3:int;
			i0 = 0, i1 = 1, i2 = 2, i3 = 3;
			while (i0 < _maxNum) {
				i1 = i0;
				i2 = ++i1;
				i3 = ++i2;
				i3++;
				
				_particleData[i0] = Math.random() * 465 >> 0;
				_particleData[i1] = Math.random() * 465 >> 0;
				_particleData[i2] = 0;
				_particleData[i3] = 0;
				
				i0 = ++i3;
			}
			
			addChild(new Stats());
			addEventListener(Event.ENTER_FRAME, onEnter);
		}

		private function onEnter(evt:Event):void {
			var gravPoint_x:Number = mouseX;
			var gravPoint_y:Number = mouseY;
			//var n:Node = _first;
			var calc:Boolean = _cnt % 4 == 0;
			
			this._bmd.lock();
			
			var i0:int, i1:int, i2:int, i3:int;
			i0 = 0;
			while (i0 < _maxNum)
			{
				i1 = i0;
				i2 = ++i1;
				i3 = ++i2;
				i3++;
				
				// 4回に1回ぐらいの処理に
				if(calc){
					var diff_x:Number = gravPoint_x - _particleData[i0];;
					var diff_y:Number = gravPoint_y - _particleData[i1];
					var acc:Number = 200 / (diff_x * diff_x + diff_y * diff_y);
					var acc_x:Number = acc * diff_x;
					var acc_y:Number = acc * diff_y;
					_particleData[i2] += acc * diff_x;
					_particleData[i3] += acc * diff_y;
				}
				
				_particleData[i0] += _particleData[i2];
				_particleData[i1] += _particleData[i3];
				
				_particleData[i2] *= 0.96;
				_particleData[i3] *= 0.96;
				
				if(calc)
				{
					if (_particleData[i0] > 465)
						_particleData[i0] -= 465;
					else if (_particleData[i0] < 0)
						_particleData[i0] += 465;
					if (_particleData[i1] > 465)
						_particleData[i1] -= 465;
					else if (_particleData[i1] < 0)
						_particleData[i1] += 465;
				}
				
				this._bmd.setPixel(_particleData[i0] >> 0, _particleData[i1] >> 0, 0xffffff);
				
				i0 = ++i3;
			}
			this._bmd.colorTransform(this._bmdRect, this._colorTransform);
			this._bmd.unlock();
			
			_cnt++;
		}
	}
}